

Icinga2 Provider

The Icinga2 provider is used to configure hosts to be monitored by Icinga2 (<https://www.icinga.com/products/icinga-2/>) servers. The provider needs to be configured with the API URL of the Icinga2 server and credentials for an API user with the appropriate permissions.

Example Usage

```
# Configure the Icinga2 provider
provider "icinga2" {
  api_url      = "https://192.168.33.5:5665/v1"
  api_user     = "root"
  api_password = "icinga"
  insecure_skip_tls_verify = true
}

# Configure a host
resource "icinga2_host" "web-server" {
  # ...
}
```

Authentication

Static credentials

Static credentials can be provided by adding an `api_user` and `api_password` in-line in the `icinga2` provider block:

Usage:

```
provider "icinga2" {
  api_url      = "https://192.168.33.5:5665/v1"
  api_user     = "root"
  api_password = "icinga"
}
```

Environment variables

You can provide your credentials via `ICINGA2_API_USER` and `ICINGA2_API_PASSWORD`, environment variables, storing your Icinga2 API user and password, respectively. `ICINGA2_API_URL` and `ICINGA2_INSECURE_SKIP_TLS_VERIFY` are also used, if applicable:

```
provider "icinga" {}
```

Usage:

```
$ export ICINGA2_API_URL=https://192.168.33.5:5665/v1
$ export ICINGA2_API_USER=root
$ export ICINGA2_API_PASSWORD=icinga
$ export ICINGA2_INSECURE_SKIP_TLS_VERIFY=true
$ terraform plan
```

Argument Reference

- `api_url` - (Required) The root API URL of an Icinga2 server. May alternatively be set via the `ICINGA2_API_URL` environment variable.
- `api_user` - (Required) The API username to use to authenticate to the Icinga2 server. May alternatively be set via the `ICINGA2_API_USER` environment variable.
- `api_password` - (Required) The password to use to authenticate to the Icinga2 server. May alternatively be set via the `ICINGA2_API_PASSWORD` environment variable.
- `insecure_skip_tls_verify` - (optional) Defaults to false. If set to true, verification of the Icinga2 server's SSL certificate is disabled. This is a security risk and should be avoided. May alternatively be set via the `ICINGA2_INSECURE_SKIP_TLS_VERIFY` environment variable.

icinga2_checkcommand

Configures an Icinga2 checkcommand resource. This allows checkcommands to be configured, updated, and deleted.

Example Usage

```
# Configure a new checkcommand on an Icinga2 Server, that can be used to monitor hosts and/or services
provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_checkcommand" "apache_status" {
  name      = "apache_status"
  templates = ["apache-status", "plugin-check-command", "plugin-check-command", "ipv4-or-ipv6"]
  command   = "/usr/lib64/nagios/plugins/check_apache_status.pl"

  arguments = {
    "-H" = "$apache_status_address$"
    "-c" = "$apache_status_critical$"
    "-p" = "$apache_status_port$"
  }
}
```

Argument Reference

The following arguments are supported:

- `arguments` - (Optional) A mapping of arguments to include with the command.
- `command` - (Required) Path to the command to be executed.
- `name` - (Required) Name by which to reference the checkcommand
- `templates` - (Optional) A list of Icinga2 templates to assign to the host.

icinga2_hostgroup

Configures an Icinga2 hostgroup resource. This allows hostgroup to be configured, updated, and deleted.

Example Usage

```
# Configure a new hostgroup to be monitored by an Icinga2 Server
provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_hostgroup" "my-hostgroup" {
  name          = "terraform-hostgroup-1"
  display_name = "Terraform Test HostGroup"
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) The name of the hostgroup.
- `display_name` - (Required) The name of the hostgroup to display in the Icinga2 interface.

icinga2_host

Configures an Icinga2 host resource. This allows hosts to be configured, updated, and deleted.

Example Usage

```
# Configure a new host to be monitored by an Icinga2 Server
provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_host" "host" {
  hostname      = "terraform-host-1"
  address       = "10.10.10.1"
  check_command = "hostalive"
  templates     = ["bp-host-web"]

  vars {
    os       = "linux"
    osver    = "1"
    allowance = "none"
  }
}
```

Argument Reference

The following arguments are supported:

- `address` - (Required) The address of the host.
- `check_command` - (Required) The name of an existing Icinga2 CheckCommand object that is used to determine if the host is available or not.
- `hostname` - (Required) The hostname of the host.
- `templates` - (Optional) A list of Icinga2 templates to assign to the host.
- `vars` - (Optional) A mapping of variables to assign to the host.

icinga2_notification

Configures an Icinga2 notification resource. This allows notifications to be configured, updated, and deleted.

Example Usage

```
# Configure a new host notification

provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_notification" "host-notification" {
  hostname = "docker-icinga2"
  command  = "mail-host-notification"
  users    = ["user"]
}
```

```
# Configure a new service notification

provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_notification" "ping-service-notification" {
  hostname      = "docker-icinga2"
  command       = "mail-service-notification"
  users         = ["user"]
  servicename   = "ping"
}
```

Argument Reference

The following arguments are supported:

- `hostname` - (Required) The hostname the notification applies to.
- `command` - (Required) Notification command to use.
- `users` - (Required) List of users to notification.
- `servicename` - (Optional) Service to send notification for.

icinga2_service

Configures an Icinga2 service resource. This allows service to be configured, updated, and deleted.

Example Usage

```
# Configure a new service to be monitored by an Icinga2 Server
provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_service" "my-service" {
  name          = "ssh"
  hostname      = "c1-mysql-1"
  check_command = "ssh"
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) The name of the Service object.
- `check_command` - (Required) The name of an existing Icinga2 CheckCommand object that is used to determine if the service is available on the host.
- `hostname` - (Required) The host to check the service's status on

icinga2_user

Configures an Icinga2 user resource. This allows users to be configured, updated, and deleted.

Example Usage

```
# Configure a new user

provider "icinga2" {
  api_url = "https://192.168.33.5:5665/v1"
}

resource "icinga2_user" "terraform" {
  name = "terraform"
  email = "terraform@dev.null"
}
```

Argument Reference

The following arguments are supported:

- `name` - (Required) The user.
- `email` - (Optional) An email string for this user. Useful for notification commands.